


A COMPUTER ADVENTURE GAME



ISLAND OF SECRETS



**STORY
CLUES & HINTS
+
COMPLETE
LISTING**

FOR COMMODORE 64
VIC 20+16K

TRS80 (32K)
APPLE

ISLAND OF SECRETS

Jenny Tyler & Les Howarth

Contents

- | | | | |
|----|--------------------------------|----|-------------------------------|
| 2 | About this book | 18 | How to play Island of Secrets |
| 4 | The story of Island of Secrets | 19 | De-bugging |
| 6 | Grandpa's map | 20 | Program listing |
| 8 | Characters you will meet | 30 | Conversion lines |
| 12 | Places you will visit | 32 | Clues |

This picture shows part of

the Library of Written Secrets.

Designed by Iain Ashman

Illustrated by Patrick Lynch

About this book

This book contains a program listing for an adventure game called "Island of Secrets" which you can type in and play on your computer. It also contains lots of exciting colour pictures which tell you the story of the game and show what some of the places and characters in it look like. These contain vital clues – look at them carefully or you won't be able to play the game.

The program listing starts on page 20. It is written to work on a Commodore 64, VIC 20 (with 16K expansion), Apple, BBC, Electron and Spectrum, though, because these computers do not all use exactly the same version of BASIC, there are changes to make depending on which one you are using. You will find out about these in the notes at the beginning of the listing.

Island of Secrets is an extremely long program to type in and even experienced programmers will find it difficult not to make mistakes. You will find some tips and hints on typing in programs on page 20.

If you don't know what an adventure game is or how to play one, you can find out more about them on the right. You can also find specific instructions for playing Island of Secrets on page 18.

How to use this book

1 Turn first to page 20 and read the tips and hints on typing in programs.


2 Check through the listing, noting how many conversions there are for your computer.

3 Type in the program slowly and carefully; if necessary stopping and saving a section part way through.

4 Now turn to page 18 and read the instructions on how to play the game.

5 Type RUN and RETURN and check that your screen display looks like the one in the first picture on page 18.


About adventure programs



You have probably heard about adventure games even if you've never played one. They were invented in the U.S.A. in 1976 and first played on mainframes with huge memories.


The player wanders round an imaginary world by typing in commands such as GO SOUTH or TAKE SWORD and the computer describes the surroundings and events as the game unfolds. Some more recent adventure games draw pictures of the characters and locations on the screen too.

The object of an adventure game is usually to collect treasure and escape from a series of dangerous situations. They are difficult to play and can take weeks, or even months, to solve.



What about Island of Secrets?

Island of Secrets is a text adventure. It doesn't show any pictures on the screen, but it does have pictures in the book instead. Unlike most adventure games, the computer will not tell you all you need to know about the world you are travelling through. There are no directions, for instance. You will need to look at the map and work out for yourself



whether you can go north, south, east or west. You will need to look at the pictures, too, to see if there are any objects that might be valuable or useful in your Quest, as the computer probably won't mention them.

This game is also different from a

Alphan's quest

To find out who Alphan is and what he has to do with the Island of Secrets you must read the Story on pages 4 and 5. His task is to find the objects of power, though he doesn't know what they are, and take them to the Island of Secrets. He must then work out what to do with them to restore light, warmth and life to the world.



traditional adventure because you play a role. You are Alphan and you must try to decide what he would do in a particular situation (based on the clues in the book) rather than what you personally would do. Other roles are played by the computer and characters sometimes move around or do surprising things which you cannot control. You are given a certain amount of strength and wisdom at the beginning of the game and these vary according to what happens to you in your role of Alphan.


If you do something wrong you may find you have lost the game. The computer will tell you you have failed and give you a score based on what you have done up to that point. You will then have to start again and try something different. Perhaps you said the wrong thing or didn't pick up something you needed.

6 You are now ready to start. Try wandering round the locations before you tackle the game seriously.

7 Have a good look at the colour section of the book to familiarize yourself with the world of Island of Secrets and its inhabitants.

8 Keep the book alongside you as you play. You will need to refer constantly to the map to find your way around.

9 You can save a game part way through if you want and load it again another time, or you can give up and get your score by typing QUIT. The book tells you how to do these things on pages 18-19.



The story of Island of Secrets

In the dim and distant past . . .

. . . the sky was blue and the grass was green. The Ancients had built The Pyramid and established communication with the Sky People. It was even agreed that one or two specially selected Earthlings should visit Sky Land.



6 Then they gave him the Cloak of Entropy which they said was essential for releasing the power of the objects and, because the task was too long for one person's lifetime, they also gave him the secret for prolonging his life.



5 It became obvious to the Sky People that a great war was inevitable, they sent Omegan back to Earth with a mission. The war would darken the Earth, they said, but when it was over Omegan must collect the secret objects and release the power that had been locked into them, thus restoring the Earth to normal.

2 The first (and last, as it turned out) Earthling Sky visitor was Omegan, youngest son of the most influential of the Ancients. He was fascinated by what he saw – the Sky People knew a lot of things Earthlings didn't.




3 Shortly after Omegan arrived, conflicts began between the Sky People and the Outer Ones. Sky messengers came down to The Pyramid and gave each of the Ancients an object, telling them to hide it carefully.



4 The Ancients wondered why the Sky People should bother to give them such ordinary looking objects. But, who were they to ask questions? They set off to find hiding places, little knowing they would never meet again.



7 The war was more catastrophic than even the Sky People had anticipated. A huge dark cloud enveloped the Earth blocking out light and heat, the ground ripped apart forming the Crimson Canyon, and the lakes were poisoned.



8 Omegan set out on his task, but it soon became apparent that it was of immense and incredible difficulty. He didn't know what he was looking for or where to look or even how many things he had to find.



9 He began to realize that the Cloak had enormous power. With its help, he built a castle for himself and assembled the secret equipment for extending his life. So what if the Earth stayed dark, he thought. No-one would find all the objects, and as for the Cloak – that was his now . . .

That was all a long time ago . . .

10 . . . but the land is still in darkness and Omegan still lives on his Island of Secrets. No-one found the objects or discovered the secret of Omegan's long life, though there are rumours of some who have tried.


11 Tales are told, for instance, of a wanderer named Median who entered the Castle of Secrets. He even drew a plan, they say, which he put in a bottle and threw into the Lake.



It could be true – who knows?

12 No-one thought Alphan would take on the task. Intelligent and wise he certainly is, but surely too slight and gentle to wrestle with the will of Omegan. True he has that map left to him by his explorer grandfather.

What if he really could bring the Earth back to normal!



Grandpa's map





Crimson Canyon

Impenetrable Forest

Suffocating Clouds

Site of Ancient University



Lake of Secrets


Blasted Land

Bluewood Stumps

Characters you will meet


Canyon Beasts

You may come across canyon beasts anywhere in the forest, though you won't be able to catch one without help.



Omegan

He moves around the castle, so you can't be sure where you will come across him.



Speaking Stone

Despite its name, this strange glassy boulder sits in dull silence.



Swampman

He lives in the stumps of the great bluewood trees, which his ancestors planted. He is a sworn enemy of the Logmen but cannot leave the swamp to get revenge or to get the liquor which he believes would restore the trees to their former glory.



The Boatman

Even the intrepid canyon beast won't go near him, so beware. It would be wise to approach him at some stage in your quest though.



The Logmen

These unfriendly characters are always on the look-out for something to entertain them. They are addicted to a lethally strong liquor which they brew themselves. Their logging activities include destroying the last of the great bluewood trees in the east.

Median

A famous scholar of Ancient times, he disappeared under mysterious circumstances.




The Scavenger

He appears to have lost his memory. Alphan will need to help him recover it somehow.



Sage of the Lilies



The Sage has meditated for so long that she has forgotten how to move anything other than her head. She has been vaguely worried for the last 500 years by a slight, but persistent irritation on her back. The lilies she guards are said to revive old memories.



Dactyl

She guards something precious. She'll need proof that you are serious about your mission.

Places you will visit




Grandpa's Chest

Grandpa kept useful things in here.



Grandpa's Shack

Alphan's grandfather built this shack as a hide-away for himself.



The Pyramid

The view from here is quite impressive. Watch your step – there's a kind of hard grey stone here.




Well of Weakness

Don't approach if you're not feeling strong. Omegan may not be far away.

Clashing Stones

If you peer closely at these you can just make out the following inscription: "Stony silence is not the thing Say stony words that you must bring"



Petrified Village

Lava engulfed this village so quickly that everything was preserved in stone. It gives you some idea of what life was like long ago.





Logmen's Cabin

Don't hang around too long in here. Stolen goods often end up on the table.



Clone Factory

Can this be the secret of Omegan's long life?



Adobe Hut

Someone appears to have lived here in recent times, but there's no sign of anyone now.

Snelm's Lair


It is said that one of the Ancients left a clue with Snelm. Look for a piece of old "paper".




Omegan's Sanctum

The end of your journey, whatever its outcome. You can't escape.

Watch your step here too.






Marble Column

This is all that remains of the University of the Ancients.

How to play Island of Secrets

1

If your screen doesn't look like this, you've probably made a typing error, so check again.




```
ISLAND OF SECRETS  TIME REMAINING:1000
YOUR STRENGTH = 100 YOUR WISDOM = 35
-----
YOU ARE ON A LEAFY PATH.
-----
LET YOUR QUEST BEGIN
WHAT WILL YOU DO?
```



When you have typed in the listing and checked it carefully, type RUN and RETURN. The word "initializing" will appear at the top of the screen, which means the computer is reading the data. After a few seconds, your screen should look like this.

2


You can use N, S, E or W instead of GO NORTH, etc.



```
ISLAND OF SECRETS  TIME REMAINING:986
YOUR STRENGTH = 97  YOUR WISDOM = 39
-----
YOU ARE BY BROKEN BRANCHES. YOU SEE
SPARKLING FRESHWATER SPRING.
-----
OK
WHAT WILL YOU DO?GET WATER
```

You are now ready to start and the computer waits for you to type in what you want to do. It only recognizes certain words and so won't understand everything you say. In most cases it needs two-word commands, such as GET WATER.


4



```
ISLAND OF SECRETS  TIME REMAINING:999
YOUR STRENGTH = 99  YOUR WISDOM = 35
-----
YOU ARE AT THE SOUTHERN EDGE OF
THE FOREST.
-----
OK
WHAT WILL YOU DO?
```

You will notice as you begin to move around that the "time remaining", shown top right of your screen, constantly decreases. The game is over when the time is up, no matter what else has happened.

5



```
ISLAND OF SECRETS  TIME REMAINING:931
YOUR STRENGTH = 63  YOUR WISDOM = 14
-----
YOU ARE BY A GROUP OF TALL TREES.
-----
OK
```

You lose the game if your strength or wisdom reach zero.

Use EAT or DRINK to increase strength.

You will also see that your strength and wisdom scores vary as the game progresses. You can increase your strength by eating or drinking, providing you have stocks of food and drink. You can only increase your wisdom by doing the right things in the game.

7


If you are about to try something risky, save the game first.



```
ISLAND OF SECRETS  TIME REMAINING:927
YOUR STRENGTH = 75  YOUR WISDOM = 30
-----
YOU ARE ON THE ROOF OF THE ANCIENT
TEMPLE.
-----
WHAT WILL YOU DO?XSAVE
```

You can stop at any point in the game and save your progress so far. Do this by typing XSAVE, as shown above. (Make sure you have your cassette recorder linked up and a tape ready first though. You will need to know how to make your computer save on tape too.)

8



```
ISLAND OF SECRETS  TIME REMAINING:1000
YOUR STRENGTH = 100 YOUR WISDOM = 35
-----
YOU ARE ON A LEAFY PATH.
-----
LET YOUR QUEST BEGIN
WHAT WILL YOU DO?XLOAD
```

When you want to continue the game, load in the game program as usual, then type XLOAD as your first instruction as shown above. You can then continue the game from where you left off.



De-bugging

3

```

ISLAND OF SECRETS  TIME REMAINING:890
-----
YOUR STRENGTH = 85  YOUR WISDOM = 27
-----
YOU ARE IN THE MIST OF THE LILIES.
-----
OK:
WHAT WILL YOU DO?GIVE AXE
GIVE THE AXE TO WHOM?

```

Sometimes you will be asked a second question, as in the situation above. You should then answer with one word. There are also special one word commands for save, load and quit, as you will see below.

6

```

ISLAND OF SECRETS INFO - ITEMS CARRIED
-----
PROVISIONS : FOOD=2      DRINK=2
-----
PRESS RETURN WHEN READY?

```

Food and drink items are just added to your food or drink score, they are not listed as melon, wine, etc.

Now type INFO and your screen will change as shown above. You can do this at any point in the game to find out what you are carrying. You start the game with two units each of food and drink and can add to this by picking up any food and drink you spot on the way.

9

```

ISLAND OF SECRETS  TIME REMAINING: 96
-----
YOUR STRENGTH = 9  YOUR WISDOM = 3
-----
YOU ARE ON A NARROW LEDGE AT THE
SUMMIT OF THE CANYON.
-----
OK:
WHAT WILL YOU DO?QUIT
YOU RELINQUISH YOUR QUEST.
YOUR FINAL SCORE=134

```

If you feel you are stuck in the middle of the game and want to give up, you can type QUIT at any time and the computer will give you your score. (Though you won't get any time bonus.)

However carefully you think you have checked a listing after you have typed it in, it is still likely that it will contain mistakes. You may not even notice them until you have played the game for a while. Here is a checklist of common problems to look out for. A good way of checking is to ask someone to read the listing out loud, including all the punctuation, while you look at the listing on the screen.

1) Have you typed in all the conversions for your computer? Check RND and clear screen in particular.

2) Brackets are very easy to miss. Check there are an equal number of opening and closing brackets in every line.

3) Quotes are also easy to miss. Make sure there is an even number of them in every line.

4) Check that you haven't missed any punctuation or confused, with; for example.

5) Have you run two lines together or accidentally split one line into two?

6) Watch out for two similar lines close together in a listing, as your eye can sometimes jump from one to the other part way through.

7) Make sure you haven't confused capital O with zero, figure 1 with capital I or figure 6 with capital G.

8) Some lines in this listing are in code. You will need to check these letter by letter to make sure they are correct.

9) The data lines have a group of 1s and 0s at the end. Make sure there are four (correct) numbers in each group.

10) Check that all the GOSUB line numbers are correct and that you haven't missed out a RETURN line.



Program listing

Here is the Island of Secrets program listing. The symbols down the left hand side indicate lines that need to be altered, added or deleted for different computers. There is a different symbol for each computer, as shown below, so make a note of the one for yours. Turn to pages 30-31 to find the new lines to type in.

Commodore 64 Δ

VIC 20 \blacktriangle

Apple \bullet

TRS-80 \blacksquare

BBC \ast

Electron \ast

Spectrum \square

It is a good idea to check through the listing before you start typing to see how many changes there are for your computer.

```

10REM ISLAND OF SECRETS
20GOSUB2820
30LET D=R:IF R=20 THEN LET D=FNR(80)
40GOSUB650:GOSUB2770 :PRINT"ISLAND OF SECRETS","TIME REMAINING:";L
50PRINT G$;TAB(0)"STRENGTH = ";INT(Y);TAB(23);"WISDOM = ";X:PRINTG$
60PRINT"YOU ARE ";I$(VAL(LEFT$(A$,1))); " ";GOSUB720:LET N=0
70FOR I=1 TO C4
80LET C=0:READ Y$
90IF L(I)=R AND F(I)<1 THEN LET C=1
100IF N>0 AND C=1 THEN LET A$=A$+" ",
110IF C=1 THEN LET A$=A$+" "+Y$:LET N=N+1
120NEXT I
130IF N>0 THEN LET A$="*YOU SEE"+A$:GOSUB720
140PRINT:PRINT G$;F$
150PRINT:PRINT"WHAT WILL YOU DO";
160INPUT E$
170LET C$="":LET X$="":LET A=0:LET O=52:LET LI=LEN(E$)
180FOR I=1 TO LI
190IF MID$(E$,I,1)=" " AND C$="" THEN LET C$=LEFT$(E$,I-1)
200IF MID$(E$,I+1,1)<>" " AND C$="" THEN LET X$=RIGHT$(E$,LI-I):LET I=LI
210NEXT I
220IF X$="" THEN LET C$=E$
230IF LEN(C$)<3 THEN LET C$=C$+"???"
240FOR I=1 TO V
250IF LEFT$(C$,3)=MID$(V$,3*(I-1)+1,3) THEN LET A=I
260NEXT I
270GOSUB760
280LET B$="":IF A=0 THEN LET A=V+1
290IF X$="???" THEN LET F$="MOST ACTIONS NEED TWO WORDS"
300IF A>V OR O=52 THEN LET F$=W$+C$+" "+X$
310IF A>V AND O=52 THEN LET F$="WHAT!"
    
```

Hints on typing in the program

You will need to go slowly and carefully to type this long listing in correctly.

Here are some hints:

1. Make sure you are sitting in a comfortable position.
2. Use a ruler or piece of blank paper to mark the line you are on, or see if you can persuade someone to read the listing out to you. (Explain that they must distinguish between TO and 2 and FOR and 4, and that they must read all the punctuation.)
3. Check each line before you press RETURN and make sure you don't miss a line altogether.
4. Stop at the bottom of each page, or more often if you like, and walk around the room for a few minutes.
5. As soon as you begin to feel irritable or panicky, STOP. You can always save what you have done and continue later.

Remember, it is easier to check each line carefully before you press return than to check through the whole listing for mistakes at the end.



If you are using a Commodore 64, leave out the LETs.

You can leave out the LETs on all computers except the Spectrum.



320LET L=L-1:LET Y=FNS(Z)

■▲△330LET B\$=STR\$(D)+STR\$(L(D))+STR\$(F(D))+STR\$(R)

□340ON INT(A/10)+1 GOSUB 590,600,610,620,630,630,640

350IF R=61 THEN LET X=X-FNR(2)+1

360IF R=14 AND FNR(3)=1 THEN LET Y=Y-1:LET F\$="YOU ARE BITTEN"

370IF F(36)<1 AND R<>F(22) THEN LET F(36)=F(36)+1:LET L(36)=R:LET Y=Y-1

380IF R<>L(16) AND L(16)>0 THEN LET L(16)=1+FNR(4)

390IF R<>L(39) THEN LET L(39)=10*(FNR(5)-1)+7+FNR(3)

400IF R=L(39) AND R<>L(43) AND F(13)>-1 THEN LET Y=Y-2:LET X=X-2

410IF R<78 THEN LET L(32)=76+FNR(2)

420IF R=33 OR R=57 OR R=73 AND FNR(2)=1 THEN LET L(25)=R

430IF R=L(32) AND FNR(2)=1 AND F(32)=0 THEN GOSUB1310

▲△440IF R=19 AND Y<70 AND F(43)=0 AND FNR(4)=1 THEN LET F\$="PUSHED INTO THE PIT":LET F(W)=1

450IF R<>L(41) THEN LET L(41)=21+(FNR(3)*10)+FNR(2)

460IF R=L(41) THEN LET F(41)=F(41)-1:IF F(41)<-4 THEN GOSUB1230

470IF F(43)=0 THEN LET L(43)=R

480IF L(43)<18 AND R<>9 AND R<>10 AND F(W-2)<1 THEN GOSUB1330

490IF R=18 THEN LET Y=Y-1

▲△500IF Y<50 THEN LET O=FNR(9):GOSUB1530:IF L(O)=R THEN LET F\$="YOU DROP SOMETHING"

510IF L<900 AND R=23 AND F(36)>0 AND FNR(3)=3 THEN GOSUB1360

520IF R=47 AND F(8)>0 THEN LET F\$=F\$+" YOU CAN GO NO FURTHER"

530IF F(8)+F(11)+F(13)=-3 THEN LET F(W)=1:GOSUB2800

540IF F(W)=0 AND L>0 AND Y>1 AND X>1 THEN GOTO30

550IF L<1 OR Y<1 THEN LET F\$="YOU HAVE FAILED, THE EVIL ONE SUCCEEDS"

560PRINT:PRINT F\$:PRINT"YOUR FINAL SCORE=";INT(X+Y+(ABS(L/7*(L<640))))

570PRINT:PRINT:PRINT"GAME OVER"

□580END

□590ON A GOSUB 810,810,810,810,810,1080,1080,1390,1530:RETURN

□600ON A-9 GOSUB 1540,1630,1670,1710,1730,1080,1760,1760,1760:RETURN

□610ON A-19 GOSUB 1820,1820,1820,1820,1910,2100,2210,2270,2270,1080:RETURN

□620ON A-29 GOSUB 2500,2500,2300,2300,2330,2350,2400,2400,2470,2540:RETURN

□630ON A-39 GOSUB 2600,2600,2720,640

640RETURN

●■▲△650LET D=D*10+LR

●■▲△660RESTORE D

●■▲△670READ A\$

□680LET D\$=RIGHT\$(A\$,4):LET A\$=LEFT\$(A\$,LEN(A\$)-4)

□690IF R=39 THEN LET D\$=MID\$("101110100",FNR(5),4)

700IF R=20 THEN LET D\$="1110"

710GOSUB2780:RETURN

□720FOR I=2 TO LEN(A\$):LET E\$=MID\$(A\$,I,1)

□730PRINT E\$;:IF E\$=" AND FNP(Z)>Z THEN PRINT

740NEXT I

750PRINT ". ";:LET A\$="":RETURN

760IF LEN(X\$)<3 THEN LET X\$=X\$+"???"

770FOR I=1 TO W

□780IF LEFT\$(X\$,3)=MID\$(Z\$,3*(I-1)+1,3) THEN LET O=I

790NEXT I:IF O=0 THEN LET O=52

800RETURN

810LET D=0:LET C=0:IF O=52 THEN LET D=A

820IF O>C4 AND O<W THEN LET D=O-C4

830IF B\$="500012"OR B\$="500053" OR B\$="500045" THEN LET D=4

If you are using a Commodore 64, leave out the spaces in the longer lines.



Take very special care here.



Take a rest now.



```

840IF B$="500070"OR B$="500037"OR B$="510011"OR B$="510041"THEN LET D=1
850IF B$="510043"OR B$="490066"OR B$="490051"THEN LET D=1
860IF B$="510060"OR B$="480056" THEN LET D=2
870IF B$="510044"OR B$="510052" THEN LET D=3
880IF B$="490051"AND F(29)=0 THEN GOSUB2110:RETURN
890IF R=L(39)AND (X+Y<180 OR R=10)THEN LET F$=W$+"LEAVE!":RETURN
900IF R=L(32)AND F(32)<1 AND D=3 THEN LET F$="HE WILL NOT LET YOU PAST":RETURN
910IF R=47 AND F(44)=0 THEN LET F$="THE ROCKS MOVE TO PREVENT YOU":RETURN
920IF R=28 AND F(7)<>1 THEN LET F$="THE ARMS HOLD YOU FAST":RETURN
930IF R=45 AND F(40)=0 AND D=4 THEN LET F$="HISSSS!":RETURN
940IF R=25 AND F(16)+L(16)<>-1 AND D=3 THEN LET F$="TOO STEEP TO CLIMB":RETURN
950IF R=51 AND D=3 THEN LET F$="THE DOOR IS BARRED!":RETURN
▲▲▲□ 960IF D>0 THEN IF MID$(D$,D,1)="0"THEN LET R=R+VAL(MID$(-10+10+01-01",D*3-2,3)):LET C=1
970LET F$="OK"
980IF D<1 OR C=0 THEN LET F$=W$+"GO THAT WAY"
▲▲ 990IF R=33 AND L(16)=0 THEN LET L(16)=FNR(4):LET F(16)=0:LET F$="THE BEAST RUNS AWAY!"
1000IF R<>L(25)OR O<>25 THEN RETURN
1010LET F$="":LET A$="#YOU BOARD THE CRAFT "
1020IF X<60 THEN LET A$=A$+S$
1030LET A$=A$+T$
1040GOSUB2740:GOSUB2760:GOSUB2760
1050IF X<60 THEN LET A$="#TO SERVE OMEGA FOREVER!":LET F(W)=1
1060IF X>59 THEN LET A$="#THE BOAT SKIMS THE DARK SILENT WATERS":LET R=57
1070GOSUB2750:GOSUB2760:GOSUB2760:RETURN
▲△1080IF((F(O)>0 AND F(O)<9)OR L(O)<>R)AND O<=C3 THEN LET F$="WHAT "+X$+"?":RETURN
1090IF B$="3450050"THEN LET Y=Y-8:LET X=X-5:LET F$="THEY ARE CURSED":RETURN
1100IF B$="3810010" THEN GOSUB1370
▲△1110IF(A=15 AND O<>20 AND O<>1)OR(A=29 AND O<>16)OR O>C3 THEN LET F$=W$+C$+" "+X$:RETURN
1120IF L(O)=R AND (F(O)<1 OR F(O)=9)AND O<C3 THEN LET L(O)=0:LET A=-1
1130IF O=16 AND L(10)<>O THEN LET L(O)=R:LET F$="IT ESCAPED":LET A=0
1140IF O>C1 AND O<C2 THEN LET F=F+2:LET A=-1
1150IF O>C2 AND O<C3 THEN LET G=G+2:LET A=-1
1160IF O>C1 AND O<C3 THEN LET L(O)=-81
1170IF A=-1 THEN LET F$="TAKEN":LET X=X+4:LET E=E+1:IF F(O)>1 THEN LET F(O)=0
1180IF B$<>"246046"OR L(11)=0 THEN RETURN
1190LET F$=U$:LET L(O)=R:IF FNR(3)<3 THEN RETURN
1200LET A$="#"+U$+R$
1210LET R=63+FNR(6):LET L(16)=1:LET F$=""
1220GOSUB2740:RETURN
1230GOSUB2770 :LET F$="":LET A$="#THE LOGMEN "+M$
1240LET F(41)=0:LET Y=Y-4:LET X=X-4
1250IF R<34 THEN LET A$=A$+"THROW YOU IN THE WATER":LET R=32
1260IF R>33 THEN LET A$=A$+"TIE YOU UP IN A STOREROOM":LET R=51
1270GOSUB2750:GOSUB2760
1280FOR I=3 TO 4
1290IF L(I)=0 THEN LET L(I)=42
1300NEXT I:RETURN
1310LET A$="*THE SWAMPMAN TELLS HIS TALE"
1320GOSUB2740:LET F(32)=-1:RETURN
1330LET F$="MEDIAN CAN DISABLE THE EQUIPMENT"
1340IF L(8)=0 THEN LET F$=F$+" AND ASKS YOU FOR THE PEBBLE YOU CARRY"
1350RETURN

```

Don't panic!



```

1360LET F(36)=-:(FNR(4)+6):LET F$="A STORM BREAKS OVERHEAD!":RETURN
1370FOR K=1 TO 30:GOSUB2770:PRINT"////LIGHTNING FLASHES!":NEXT K
1380LET L(39)=R:LET Y=Y-8:LET X=X-2:RETURN
1390IF (O<>24 AND L(O)>0) OR O=52 THEN LET F$="YOU DON'T HAVE THE "+X$:RETURN
▲1400PRINT"GIVE THE ";X$;" TO WHOM";:INPUT X$
1410LET Q=0:GOSUB760:LET N=0:LET O=Q
1420IF R<>L(N) THEN LET F$="THE "+X$+" IS NOT HERE":RETURN
▲△1430IF B$="10045" AND N=40 THEN LET L(O)=81:LET F(40)=1:LET F$="THE SNAKE UNCURLS"
▲△1440IF B$="2413075"AND N=30 AND G>1 THEN LET F(11)=0:LET F$="HE OFFERS HIS STAFF":LET G=G-1
□1450LET B$=LEFT$(B$,3):LET F$="IT IS REFUSED"
1460IF B$="300"AND N=42 THEN LET X=X+10:LET L(O)=81
1470IF B$="120"AND N=42 THEN LET X=X+10:LET L(O)=81
■▲△1480IF B$="40-"AND N=32 THEN LET F(N)=1:LET L(O)=81
□1490IF LEFT$(B$,2)="80"AND N=43 THEN LET L(O)=81:GOSUB1560
1500IF L(O)=81 OR (O=24 AND L(11)>0 AND G>0)THEN LET F$="IT IS ACCEPTED"
1510IF N=41 THEN LET L(O)=51:LET F$="IT IS TAKEN"
1520RETURN
1530IF O=4 AND L(O)=0THEN LET L(O)=81:LET X=X-1:LET F$="IT BREAKS!":RETURN
1540IF L(O)=0 AND O<C1 THEN LET L(O)=R:LET F$="DONE":LET E=E-1
1550RETURN
1560LET A$="*HE TAKES IT ":IF R<>8 THEN LET A$=A$+"RUNS DOWN THE CORRIDOR,"
▲△1570GOSUB2740:LET A$="*AND CASTS IT INTO THE CHEMICAL VATS, PURIFYING THEM WITH"
▲△1580LET A$=A$+" A CLEAR BLUE LIGHT REACHING FAR INTO THE LAKES AND RIVERS BEYOND"
1590LET F(8)=-1:GOSUB2750:GOSUB2760:GOSUB2760:RETURN
1600IF L(I)<>0 AND I<C1 THEN LET I=I+1:GOTO1600
1610IF L(I)=0 THEN LET L(I)=R:LET F(I)=0:GOSUB1540:LET F$="YOU DROP SOMETHING"
1620RETURN
1630IF(O<C1 OR O>C3) AND X$<>"???" THEN LET F$=W$+C$+" "+X$:LET X=X-1:RETURN
1640LET F$="YOU HAVE NO FOOD":IF F>0 THEN LET F=F-1:LET Y=Y+10:LET F$="OK"
1650IF O=3THEN LET X=X-5:LET Y=Y-2:LET F$="THEY MAKE YOU VERY ILL!"
1660RETURN
1670IF O=31 THEN GOSUB2380:RETURN
1680IF X$<>"???" AND (O<21 OR O>C3) THEN LET F$=W$+C$+" "+X$:LET X=X-1:RETURN
1690LET F$="YOU HAVE NO DRINK":IF G>0 THEN LET G=G-1:LET Y=Y+7:LET F$="OK"
1700RETURN
□1710IF LEFT$(B$,4)="1600" THEN LET F(O)=-1:LET F$="IT ALLOWS YOU TO RIDE"
1720RETURN
1730IF B$="2644044" THEN LET F$="CHEST OPEN":LET F(6)=9:LET F(5)=9:LET F(15)=9
1740IF B$="2951151" THEN LET F$="THE TRAPDOOR CREAKS":LET F(29)=0:LET X=X+3
1750RETURN
1760LET Y=Y-2:IF B$="3577077"AND L(9)=0THEN LET F(23)=0:LET L(23)=R
1770IF V>15 AND V<19 AND (L(9)=0 OR L(15)=0) THEN LET F$="OK"
▲△1780IF B$="1258158"OR B$="2758158"AND L(15)=0 THEN LET F(12)=0:LET F(27)=0:LET F$="CRACK!"
□1790IF LEFT$(B$,4)="1100" AND R=10 THEN GOSUB1980
1800IF A=18 AND (O>29 AND O<34) OR (O>38 AND O<44) OR O=16 THEN GOSUB1900
1810RETURN
1820LET Y=Y-2:LET X=X-2:IF R<>L(O)AND L(O)<>0THEN RETURN
1830IF O=39 THEN LET F$="HE LAUGHS DANGEROUSLY"
1840IF O=32 THEN LET F$="THE SWAMPMAN IS UNMOVED"
1850IF O=33 THEN LET F$=W$+"TOUCH HER!":LET L(3)=81
1860IF O=41 THEN LET F$="THEY THINK THAT'S FUNNY!"
1870IF R=46 THEN GOSUB1200

```

There are lots of capital Os and zeros here - don't confuse them.



Stop now and walk round the room.



```

□ 1880IF LEFT$(B$,4)="1400"AND R=L(39)THEN GOSUB1980
1890LET Y=Y-8:LET X=X-5:RETURN
1900IF L(9)>0 THEN RETURN
1910LET Y=Y-12:LET X=X-10:LET F$="THAT WOULD BE UNWISE!"
1920IF R<>L(0) THEN RETURN
1930LET F(W)=1:LET A$="#THUNDER SPLITS THE SKY!":LET F$=""
1940LET A$=A$+"IT IS THE TRIUMPHANT VOICE OF OMEGAN.":GOSUB2740
1950LET A$="#WELL DONE ALPHAN! THE MEANS BECOME THE END.."
1960LET A$=A$+"I CLAIM YOU AS MY OWN! HA HA HAH!":GOSUB2750
1970GOSUB2760:LET X=0:LET L=0:LET Y=0:RETURN
□ 1980GOSUB2770 :ON 0-10 GOSUB2010,2060,2060,2060
1990LET X=X+10:LET L(0)=81:LET F(0)=-1:GOSUB720:GOSUB2760:GOSUB2760
2000RETURN
2010LET A$="#IT SHATTERS RELEASING A DAZZLING RAINBOW OF COLOURS!"
2020IF L(2)<>R THEN RETURN
2030LET A$=A$+"THE EGG HATCHES INTO A BABY DAKTYL "+D$
2040LET L(39)=81:LET L(2)=81:LET F(2)=-1:LET Y=Y+40
2050RETURN
2060IF L(13)<>R THEN RETURN
2070LET A$="*THE COAL BURNS WITH A WARM RED FLAME":LET F(13)=-1
▲△ 2080IF R=10 AND R=L(39)THEN LET A$=A$+" WHICH DISSOLVES OMEGAN'S CLOAK":LET Y=Y+20
2090RETURN
2100IF R<>51 OR F(29)>0THEN LET F$=W$+C$+" HERE":LET X=X+1
2110LET X=X-1:LET R=FNR(5):GOSUB2770 :PRINT"SWIMMING IN THE POISONOUS WATERS"
2120LET J=0:LET B$="":LET F$="YOU SURFACE":PRINT"YOUR STRENGTH = ";INT(Y)
2130FOR I=1 TO R
2140IF Y<15 THEN PRINT"YOU ARE VERY WEAK"
□ 2150PRINT"WHICH WAY";INPUT X$:LET X$=LEFT$(X$,1):LET B$=B$+X$:NEXT I
2160FOR I=1TO R
□ 2170LET Y=FNS(Z)-3:IF MID$(B$,I,1)="N" THEN LET J=J+1
2180NEXT I:IF R/2>J AND Y>1 THEN GOTO 2110
2190IF Y<2 THEN LET F$="YOU GOT LOST AND DROWNED"
2200LET R=30+FNR(3):RETURN
2210IF F(36)>-1 THEN RETURN
2220GOSUB2770 :PRINT"YOU CAN RUN TO SHELTER IN:":PRINT"1) GRANDPA'S SHACK"
▲△ 2230PRINT"2) CAVE OF SNELM":PRINT"3) LOG CABIN":PRINT"CHOOSE FROM 1-3":INPUT A$
□ 2240IF A$>"0"AND A$<"4"THEN LET R=ASC(MID$("A",A$),1)-21:LET F(22)=-R
2250PRINT"YOU RUN BLINDLY THROUGH THE STORM":LET F$="YOU REACH SHELTER"
2260GOSUB2760:RETURN
2270IF B$="3075075"OR B$="3371071"THEN LET F$="HOW WILL YOU DO THAT"
▲△ 2280IF B$="3371071" AND A=28 THEN LET F(3)=0:LET F$="SHE NODS SLOWLY":LET X=X+5
2290RETURN
2300LET F$="EXAMINE THE BOOK FOR CLUES"
□ 2310IF LEFT$(B$,3)="600"THEN LET F$=L$
2320RETURN
2330IF B$="40041" THEN LET F(4)=-1:LET F$="FILLED"
2340RETURN
2350LET F$=X$:IF X$=H$ AND R=47 AND F(8)=0THEN LET F(44)=1:LET F$=J$
2360IF X$<P$ OR R<>L(42) OR L(3)<81 OR L(12)<81 THEN RETURN
2370LET F$="HE EATS THE FLOWERS- AND CHANGES":LET F(42)=1:LET F(43)=0:RETURN
2380IF F(4)+L(4)<>-1 THEN LET F$="YOU DON'T HAVE "+X$:RETURN
▲△ 2390GOSUB2770 :PRINT"YOU TASTE A DROP AND..":GOSUB2760:LET F$="*OUCH!":LET Y=Y-4:LET X=X-7

```

You're half way through.



Note there is a space here.



```

2400GOSUB2770:FOR I=1 TO ABS(F(36))+3
2410LET L=L-1:IF Y<100 OR -R=F(22) THEN LET Y=Y+1
2420PRINT"TIME PASSES":GOSUB2760
2430NEXT I
2440IF L>100 OR F(36)<1 THEN LET X=X+2:LET F(36)=1
2450IF A=37 OR A=36 THEN LET F$="OK"
2460RETURN
2470IF R=L(25)THEN LET F$="THE BOATMAN WAVES BACK"
□ 2480IF LEFT$(B$,3)="700"THEN LET F(7)=1:LET F$=N$:LET X=X+B
2490RETURN
□ 2500LET F$="A-DUB-DUB":IF LEFT$(B$,4)<>"2B15"THEN RETURN
2510IF F(0)=1 THEN LET F(0)=0:LET F$=K$:RETURN
2520IF L(5)=0 THEN LET F(8)=0:GOSUB1080:LET F$="THE STONE UTTERS "+H$
2530RETURN
2540GOSUB2770 :PRINT" INFO - ITEMS CARRIED":GOSUB2780
● 2550PRINT G$:TAB(0);" FOOD=";F:TAB(23);"DRINK=";G:PRINT G$:LET F$="OK"

```

Take another rest soon.



```

2560FOR I=1 TO C4
2570READ Y$:IF L(I)=0 THEN PRINT Y$
2580NEXT I
2590PRINTG$:GOSUB2730:RETURN
2600LET C$="LOAD":IF A=41 THEN LET C$="SAVE"
2610PRINT"PREPARE TO ";C$:GOSUB2730
*□●■ 2620IF A=40 THEN OPEN 1,1,0,"ISDATA"
*□●■ 2630IF A=41 THEN OPEN 1,1,1,"ISDATA"
▲△ 2640IF A=41 THEN LET F(50)=R:LET F(49)=Y:LET F(48)=X:LET F(47)=F:LET F(46)=G:LET F(45)=L
□ 2650FOR I=1 TO W
*□●■ 2660IF A=40 THEN INPUT#1,L(I):INPUT#1,F(I)
*□●■ 2670IF A=41 THEN PRINT#1,L(I):PRINT#1,F(I)
□ 2680NEXT I
*□●■ 2690CLOSE 1

```

```

▲△ 2700IF A=40 THEN LET R=F(50):LET Y=F(49):LET X=F(48):LET F=F(47):LET G=F(46):LET L=F(45)
2710LET F$="OK":RETURN
2720LET F(W)=-1:LET F$="YOU RELINQUISH YOUR QUEST.":LET L=1:RETURN
2730INPUT"PRESS RETURN";A$:RETURN
2740GOSUB2770
2750GOSUB720:GOSUB2760:RETURN

```

You can stop part way through and save what you have typed so far on to cassette.



```

□●■▲△ 2760FOR D=1 TO 2000:NEXT D:RETURN
●▲△ 2770CLS:RETURN
●■▲△ 2780RESTORE LR+810
2790RETURN
2800LET A$="*THE WORLD LIVES WITH NEW HOPE!":GOSUB2750
2810LET F$="YOUR QUEST IS OVER":RETURN
2820PRINT"INITIALISING"
□●■▲△ 2830LET LR=2860:LET Z=39:LET Z=INT(Z*.8)
2840LET V=42:LET W=51:LET C4=43
□ 2850DIM I$(7)
□ 2860DIM L(52),F(52)

```

Don't forget to load what you have already typed, though, before you carry on to the next bit.

```

2870DATA"4THE FURTHEST DEPTHS OF THE FOREST1001"
2880DATA"4THE DEPTHS OF THE MUTANT FOREST1000"
2890DATA"7A PATH OUT OF THE OVERGROWN DEPTHS1000"
2900DATA"6A CARNIVOROUS TREE1000"
2910DATA"4A CORRAL BENEATH THE CRIMSON CANYON1110"

```

2920DATA"7THE TOP OF A STEEP CLIFF1011"
2930DATA"4THE MARSH FACTORY1001"
2940DATA"4THE SLUDGE FERMENTATION VATS1110"
2950DATA"7THE UPPERMOST BATTLEMENTS1001"
2960DATA"4OMEGAN'S SANCTUM1110"
2970DATA"4SNELM'S LAIRO0001"
2980DATA"2A DARK CAVE0000"
2990DATA"1BROKEN BRANCHES0100"
3000DATA"1A THICKET OF BITING BUSHES0000"
3010DATA"1A HUGE GLASSY STONE1110"
3020DATA"7THE EDGE OF CRIMSON CANYON0011"
3030DATA"4THE CLONE FACTORY0101"
3040DATA"4A CORRIDOR OF CLONE STORAGE CASKS1100"
3050DATA"7EDGE OF THE WELL0000"
3060DATA"4THE ROOM OF SECRET VISIONS1110"
3070DATA"4SNELM'S INNER CHAMBER0111"
3080DATA"3THE SOUTHERN EDGE OF THE FOREST0101"
3090DATA"7A LEAFY PATH1000"
3100DATA"3A FORK IN THE PATH0100"
3110DATA"7AN APPARENTLY UNCLIMBABLE ROCKY PATH1100"
3120DATA"7A LEDGE ATOP THE CRIMSON CANYON0010"
3130DATA"4A TALL ENTRANCE CHAMBER1101"
3140DATA"4A LOW PASSAGE WITH ARMS REACHING FROM THE WALLS1010"
3150DATA"7THE APPROACH TO THE WELL OF DESPAIR0001"
3160DATA"4A DIM CORRIDOR DEEP IN THE CASTLE1010"
3170DATA"4THE STAGNANT WATERS OF THE CRAWLING CREEK1001"
3180DATA"4A SHALLOW POOL OFF THE CREEK1100"
3190DATA"7A LOG PIER, JUTTING OUT OVER THE CREEK0000"
3200DATA"4A STRETCH OF FEATURELESS DUNES1100"
3210DATA"1A GROUP OF TALL TREES1010"
3220DATA"7A NARROW LEDGE AT THE SUMMIT OF THE CANYON0011"
3230DATA"2A MONSTEROUS PORTAL IN THE CASTLE WALL0011"
3240DATA"4A CHAMBER INCHES DEEP WITH DUST0001"
3250DATA"4HERE1111"
3260DATA"2A CARVED ARCHWAY0010"
3270DATA"4A SMALL HUT IN THE LOG SETTLEMENT0111"
3280DATA"1A HUGE SPLIT-LOG TABLE1001"
3290DATA"4THE PORCH OF THE LOGMEN'S CABIN0110"
3300DATA"4GRANDPA'S SHACK1101"
3310DATA"3A CLEARING IN THE TREES BY A RICKETY SHACK0010"
3320DATA"4THE NEST OF A HUGE DACTYL0111"
3330DATA"6THE CASTLE OF DARK SECRETS BY TWO HUGE STONES0011"
3340DATA"4A ROOM LITTERED WITH BONES0111"
3350DATA"4THE CELL OF WHISPERED SECRETS0111"
3360DATA"4THE LIBRARY OF WRITTEN SECRETS0111"
3370DATA"4A REFUSE STREWN STOREROOM1111"
3380DATA"4THE LOGMEN'S HALL0000"
3390DATA"5A LOG BUILDING1000"
3400DATA"7A RUTTED HILLSIDE1100"
3410DATA"7A WINDSWEPT PLAIN AMONGST STONE MEGALITHS0100"
3420DATA"7THE STEPS OF AN ANCIENT PYRAMID1010"
3430DATA"7THE ISLAND OF SECRETS0111"

Make sure you
have four
numbers at the
end of each of
these lines.



3440DATA"1A BROKEN MARBLE COLUMN1001"
3450DATA"7AN EXPANSE OF CRACKED, BAKED EARTH1100"
3460DATA"4A DESERTED ADOBE HUT1010"
3470DATA"4A LIVID GROWTH OF MAD ORCHIDS1011"
3480DATA"4A CORNER STREWN WITH BROKEN CHAIRS0111"
3490DATA"7THE BRIDGE NEAR TO A LOG SETTLEMENT0011"
3500DATA"1A CRUMBLING MASS OF PETRIFIED TREES1011"
3510DATA"3THE EDGE OF THE PYRAMID1101"
3520DATA"7THE ROOF OF THE ANCIENT PYRAMID0100"
3530DATA"3AN IMPASSABLE SPLIT IN THE PYRAMID1110"
3540DATA"7A BARREN BLASTED WASTELAND0001"
3550DATA"4AN EXPANSE OF BLEAK, BURNT LAND1100"
3560DATA"5A DELAPIDATED ADOBE HUT0110"
3570DATA"4THE HEART OF THE LILIES0101"
3580DATA"4THE MIDST OF THE LILIES1100"
3590DATA"3A RIVER'S EDGE BY A LOG BRIDGE0100"
3600DATA"3A PETRIFIED VILLAGE BY A RIVER CROWDED WITH LILIES0100"
3610DATA"4THE REMAINS OF A VILLAGE1100"
3620DATA"3THE ENTRANCE TO A PETRIFIED VILLAGE1100"
3630DATA"4A SWAMP MATTED WITH FIBROUS ROOTS1100"
3640DATA"2A VILLAGE OF HOLLOW STUMPS DEFYING THE SWAMP0100"
3650DATA"4A TUNNEL INTO ONE OF THE TREE STUMPS1100"
3660DATA"4A HOLLOW CHAMBER MANY METRES IN DIAMETER1110"
3670DATA"A SHINY APPLE"
3680DATA"A FOSSILISED EGG"
3690DATA"A LILY FLOWER"
3700DATA"AN EARTHENWARE JUG"
3710DATA"A DIRTY OLD RAG"
3720DATA"A RAGGED PARCHMENT"
3730DATA"A FLICKERING TORCH"
3740DATA"A GLISTENING PEBBLE"
3750DATA"A WOODMAN'S AXE"
3760DATA"A COIL OF ROPE"
3770DATA"A RUGGED STAFF"
3780DATA"A CHIP OF MARBLE"
3790DATA"A POLISHED COAL"
3800DATA"A PIECE OF FLINT"
3810DATA"A GEOLOGIST'S HAMMER"
3820DATA"A WILD CANYON BEAST"
3830DATA"A GRAIN LOAF"
3840DATA"A JUICY MELON"
3850DATA"SOME BISCUITS"
3860DATA"A GROWTH OF MUSHROOMS"
3870DATA"A BOTTLE OF WATER"
3880DATA"A FLAGON OF WINE"
3890DATA"A FLOWING SAP"
3900DATA"A SPARKLING FRESHWATER SPRING"
3910DATA"THE BOATMAN"
3920DATA"A STRAPPED OAK CHEST"
3930DATA"A FRACTURE IN THE COLUMN"
3940DATA"A MOUTH-LIKE OPENING"
3950DATA"AN OPEN TRAPDOOR"

Only one more
page to go!



Conversion lines

Commodore 64 ▲

Leave out the LETs in lines 440, 500
960, 990, 1080, 1110, 1430, 1440, 1570,
1580, 1780, 2080, 2280, 2390, 2640,
2700, 4340

Leave out the spaces (except where
marked in quotes in the text) in lines
440, 960, 1110, 1440, 2230, 2390

eg. 1440IFB\$="2413075"ANDN=30ANDG>1THENF(11)=0:F\$="HE OFFERS HIS STAFF":G=G-1

In line 4170, use SHIFT and the letter
to get lower case letters.

Replace \ with pound sign.

```
330 GOSUB 4500
650 RESTORE
660 FOR I=1 TO D:READ A$:NEXT I
670 Leave out this line
1480 IF B$="40" AND F(4)<0 AND N=32 THEN LET F(N)=1:LET L(0)=81
2760 FOR D=1 TO 900:NEXT D:RETURN
2770 PRINT CHR$(147):RETURN
2780 RESTORE:FOR I=1 TO 80:READ TP$:NEXT I
2830 LET Z=31
4110 FOR I=1 TO 43:READ TP$:NEXT I
4190 LET G$="-----"
4405 IF L(I)>127 THEN LET L(I)=L(I)-96
4440 DEF FNP(Z)=POS(0)
4500 B$=STR$(0)+STR$(L(0))+STR$(F(0))+STR$(R)
4510 B$=STR$(VAL(B$)):B$=RIGHT$(B$,LEN(B$)-1)
4520 RETURN
```

Vic 20 ▲

Use C64 conversion lines plus the lines listed below.

```
50 PRINT G$;TAB(0)"STRENGTH = ";INT(Y):PRINT "WISDOM = ";X:PRINT G$
1400 PRINT "GIVE THE ";X$;" TO WHOM":INPUT X$
2830 LET Z=14
4190 LET G$="-----"
```

BBC and Electron★

```
2620IF A=40 THEN FL=OPENIN "ISDATA"
2630IF A=41 THEN FL=OPENOUT "ISDATA"
2660IF A=40 THEN INPUT#FL,L(I),F(I)
2670IF A=41 THEN PRINT#FL,L(I),F(I)
2690CLOSE#FL
```

```

Spectrum □ 50 PRINT G$;TAB(0);"STRENGTH = ";INT(Y);TAB(18);"WISDOM = ";X;PRINT G$
55 LET LI=VAL(A$(1))
60 PRINT "YOU ARE ";I$(LI)( TO H(LI));" ";GOSUB 720:LET N=0
160 INPUT E$:PRINT " ";E$
180 FOR I=1 TO LI-1
190 IF E$(I)=" " AND C$="" THEN LET C$=E$(TO I-1)
200 IF E$(I+1)<>" " AND C$>"" THEN LET X$=E$(I+1 TO );LET I=LI-1
250 IF C$( TO 3)=V$(3*(I-1)+1 TO 3*(I-1)+3) THEN LET A=I
340 GOSUB G(A)
580 STOP
590-630 Leave out these lines
680 LET D$=A$(LEN(A$)-3 TO );LET A$=A$(TO LEN(A$)-4)
690 IF R=39 THEN LET RD=FNR(5):LET D$="101110100"(RD TO RD+3)
720 FOR I=2 TO LEN(A$):LET E$=A$(I)
730 PRINT E$;IF E$=" " AND PEEK(23688)<Z THEN PRINT
780 IF X$( TO 3)=Z$(3*(I-1)+1 TO 3*(I-1)+3) THEN LET Q=I
955 IF D=5 THEN GOTO 970
960 IF D>0 THEN IF D$(D)="0" THEN LET R=R+VAL("-10+10+01-01"(D*3-2 TO D*3)):LET C=1
1450 LET B$=B$( TO 3):LET F$="IT IS REFUSED"
1490 IF B$( TO 2)="80" AND N=43 THEN LET L(D)=81:GOSUB 1560
1710 IF B$( TO 4)="1600" THEN LET F(0)=-1:LET F$="IT ALLOWS YOU TO RIDE"
1790 IF B$( TO 4)="1100" AND R=10 THEN GOSUB 1980
1880 IF B$( TO 4)="1400" AND R=L(39) THEN GOSUB 1980
1980 GOSUB 2770:GOSUB 2010*((D-10)=1)+2060*((D-10)>1)
2150 PRINT "WHICH WAY ";:INPUT X$:PRINT X$:LET X$=X$(1):LET B$=B$+X$:NEXT I
2170 LET Y=FNS(Z)-3:IF B$(I)="N" THEN LET J=J+1
2240 IF A$>"0" AND A$<"4" THEN LET R=CODE("A >")(VAL(A$))-21:LET F(22)=-R
2310 IF B$( TO 3)="600" THEN LET F$=L$
2480 IF B$( TO 3)="700" THEN LET F(7)=1:LET F$=N$:LET X=X+8
2500 LET F$="A-DUB-DUB":IF B$( TO 4)<>"2815" THEN RETURN
2620 REM
2630 REM
2650 REM
2660 IF A=40 THEN LOAD "ISDATA" DATA L():LOAD "ISDATA" DATA F()
2670 IF A=41 THEN SAVE "ISDATA" DATA L():SAVE "ISDATA" DATA F()
2680 REM
2690 REM
2760 FOR D=1 TO 70:NEXT D:RETURN
2830 LET LR=2860:LET Z=8
2850 DIM I$(7,7)
2860 DIM L(52):DIM F(52):DIM H(7):DIM G(43)
4105 DATA 2,6,2,2,7,7,2
4125 FOR I=1 TO 7:READ H(I):NEXT I
4190 LET G$="-----"
4400 LET L(I)=CODE(H$(I))-32:LET F(I)=CODE(Q$(I))-48
4430 DEF FNR(Z)=INT(RND*Z+1)
4440 FOR I=1 TO 43:READ G(I):NEXT I
4470 DATA 810,810,810,810,810,1080,1080,1390,1530
4480 DATA 1540,1630,1670,1710,1730,1080,1760,1760,1760,1760
4490 DATA 1820,1820,1820,1820,1910,2100,2210,2270,2270,1080
4500 DATA 2500,2500,2300,2300,2330,2350,2400,2400,2470,2540
4510 DATA 2600,2600,2720,640

```

Line 4170 - use SHIFT / 0 to get reverse letters and use these instead of lower case.

To get \ use SHIFT / CLEAR

```

5 CLEAR 1000
50 PRINT G$;TAB(0);"STRENGTH = ";INT(Y);TAB(18);"WISDOM = ";X:PRINT G$
330 GOSUB 4500
650 RESTORE
660 FOR I=1 TO D:READ A$:NEXT I
670 Leave out this line
1480 IF B$="40" AND F(4)<0 AND N=32 THEN LET F(N)=1:LET L(0)=81
2620 IF A=40 THEN OPEN "I",#-1,"ISDATA"
2630 IF A=41 THEN OPEN "O",#-1,"ISDATA"
2660 IF A=40 THEN INPUT#-1,L(I),F(I)
2670 IF A=41 THEN PRINT#-1,L(I),F(I)
2690 CLOSE#-1
2760 FOR D=1 TO 900:NEXT D:RETURN
2780 RESTORE:FOR I=1 TO 80:READ TP$:NEXT I
2830 LET Z=24
4110 FOR I=1 TO 43:READ TP$:NEXT I
4190 LET G$="-----"
4430 DEF FNR(Z)=INT(RND(0)*Z+1)
4440 DEF FNP(Z)=POS(0)
4500 LET B$=STR$(0)+STR$(L(0))+STR$(F(0))+STR$(R)
4510 LET B$=STR$(VAL(B$)):LET B$=RIGHT$(B$,LEN(B$)-1)
4520 RETURN

```

Apple ●

```

40 GOSUB 650:GOSUB 2770:PRINT "ISLAND OF SECRETS":PRINT "TIME REMAINING:":L
50 PRINT G$;TAB(1);"STRENGTH = ";INT(Y);TAB(23);"WISDOM = ";X:PRINT G$
650 RESTORE
660 FOR I=1 TO D:READ A$:NEXT I
670 Leave out this line
2550 PRINT G$;TAB(1);"FOOD=";F;TAB(23);"DRINK=";G:PRINT G$;LET F$="OK"
2620 IF A=40 THEN PRINT FL$;"OPEN ISDATA":PRINT FL$;"READ ISDATA"
2630 IF A=41 THEN PRINT FL$;"OPEN ISDATA":PRINT FL$;"WRITE ISDATA"
2660 IF A=40 THEN INPUT L(I):INPUT F(I)
2670 IF A=41 THEN PRINT L(I):PRINT F(I)
2690 PRINT FL$;"CLOSE ISDATA"
2760 FOR D=1 TO 900:NEXT D:RETURN
2770 HOME:RETURN
2780 RESTORE:FOR I=1 TO 80:READ TP$:NEXT I
2830 LET Z=31
4110 FOR I=1 TO 43:READ TP$:NEXT I
4170 Leave out this line
4400 READ L(I):LET F(I)=ASC(MID$(Q$,I,1))-48
4440 DEF FNP(Z)=POS(0)
4455 FL$=CHR$(4)
4470 DATA 45,46,71,41,44,21,27,15,53,62,75,58,80,67,44,5,44,42,60,21,44,42,77
4480 DATA 13,33,44,58,15,51,75,41,78,71,50,77,23,19,10,10,45,42,38,38,0,0
4490 DATA 0,0,0,0,0,0,0,0

```

Clues

Don't look at this page unless you are really stuck. It contains clues to some of the puzzles in *Island of Secrets*. If you are still completely stuck after studying these carefully, write to the address below, enclosing a stamped addressed envelope, and you will be sent some further clues. Mark your envelope ISLAND OF SECRETS.

If the program does not work or you get error messages, check each line very carefully for mistakes, using the checklist on debugging on page 19. If you still have problems, send a copy on cassette to the address below stating what computer you are using and what the problem is. Mark the envelope ISLAND OF SECRETS and enclose stamps for return postage.

Here is a list of words the computer understands.

EAST WEST UP DOWN IN OUT
SCAVENGER MEDIAN NORTH SOUTH
OMEGAN SNAKE LOGMEN
BOOKS ROOTS WRAITHS CLOAK
VILLAGER LIQUOR SWAMPMAN SAGE
COLUMN STONE TRAPDOOR
WATER WINE SAIL BOAT CHEST
MELON BISCUITS MUSHROOMS
HAMMER CANOYON BEST LOAF
AXE ROPE STAFF COAL FLINT
RAG PARCHEMENT TORCH PEBBLE
INFO QUIT APPLE EGG FLOWERS JUG
EXAMINE FILL SAY WAIT REST WAVE
SCRATCH CATCH RUB POLISH READ
HIT KILL SWIM SHELTER HELP
TAP BREAK FIGHT STRIKE ATTACK
DRINK RIDE OPEN PICK CHOP CHIP
GET TAKE GO GIVE DROP LEAVE EAT

Snelm's inner chamber

Notice the scrap of parchment here. You need to have opened grandfather's chest to read it. It must be a clue!

First published in 1984 by
Usborne Publishing Ltd
20 Garrick Street
London WC2E 9BJ

Copyright © 1984 Usborne Publishing Ltd

Dactyl's nest

Dactyl needs proof that you are serious.
Visit the petrified village first.

The "unclimbable" path

The beast can climb up here. Perhaps it will take you. You will need something to catch it with though.

The logmen's settlement

If the logmen catch you, they will take things from you. You can get them back though. Look on the split log table.

The lilies

Perhaps you could do something about the Sage's itch. The flowers have memory-restoring powers so take some if you can.

Omegan's sanctum

Take a good look at the floor. It may help you work out what to do. You need to have found all five objects of power to be successful.

The Scavenger

There's a very close connection between him and Median. You could try to bring back his memory. Give him two things and say something too.

The Speaking Stone

A rag will be useful here. Keep trying if nothing much happens first time.

Marble column

This might jog Median's memory. Chip a bit off and see.

The Pyramid

There's a piece of flint up here.

The blue wood stumps

There's something here, but you'll need a gift for the swampman to get it.

The name Usborne and the device  are trade marks of Usborne Publishing Ltd. All rights reserved. No part of this publication may be reproduced, stored in a retrieval system or transmitted in any form or by any means, electronic, mechanical, photocopying, recording or otherwise, without the prior knowledge of the publisher.

COMPUTER ADVENTURES



The first two books in this series mark the beginning of a completely new kind of adventure game. They each contain an exciting adventure program for you to type into your computer. As you play the game on the computer, the imaginary world of the game is vividly recreated in colorful pictures in the book and you can find clues hidden in the pictures to help you.

An adventure game takes you into a different world where you have to pit your wits against magical forces, evil creatures and powerful tyrants. The books give thorough instructions on how to play and hints on what to do if you get stuck.

The programs are written in a standard BASIC and there are conversions listed to make them suitable for the Commodore 64, expanded VIC 20, TRS-80 Color Computer (32K) and Apple II.

Highgate™ is the
Trade Mark of
Usborne Publishing Ltd

ISBN 1 85123 045 9